Pensieve Header: Tangle closures in the β-framework.

```
<< KnotTheory`
GD[K_] := GD @@ (
    PD[K] /. X[i_, j_, k_, l_] :> If[PositiveQ[X[i, j, k, l]],
        Ar[l, i, +1], Ar[j, i, -1]
      ]
  )
```

```
Loading KnotTheory` version of August 22, 2010, 13:36:57.55.
Read more at http://katlas.org/wiki/KnotTheory.
```

```
Draw[expr_] := expr /. gd_GD :> Draw[gd];
Draw[gd_GD] := Module[
    {n = Length[gd], h, k = 0},
    Graphics[{
      Line[{{0, 0}, {2 n + 1, 0}}],
      Table[Text[i, {i, -0.3}], {i, 2 n}],
      (List @@ gd) /. {
        Ar[i_, j_, s_] :> {
          h = Abs[i - j] / 2;
          BezierCurve[{
            {i, 0}, {i, h}, {(i + j) / 2, h}, {j, h}, {j, 0}
           }, SplineDegree -> 2],
          Text[s * (++k), {(i + j) / 2, h - 0.3}],
          Line[{{j - 0.2, 0.4}, {j, 0}, {j + 0.2, 0.4}}]
         }
       }
     }
    ]
  ];
Draw[GD[Knot[8, 17]]]
```

KnotTheory::loading : Loading precomputed data in PD4Knots`.



```
βSimplify = Factor;
SetAttributes[βCollect, Listable];
βCollect[B[ω_, μ_]] := B[
   βSimplify[ω],
   Collect[μ, _h, Collect[#, _t, βSimplify] &]
  ];
(* "L" for "Labels" *)
hL[β_] := Union[Cases[β, h[s_] :> s, Infinity]];
tL[β_] := Union[Cases[β, t[s_] | T_s_ :> s, Infinity]];
dL[β_] := Union[hL[β], tL[β]];
SetAttributes[βForm, Listable];
βForm[B[ω_, μ_]] := Module[
   {tails, heads, mat},
   tails = tL[B[ω, μ]]; heads = hL[B[ω, μ]];
   mat = Outer[βSimplify[Coefficient[μ, h[#1] t[#2]]] &, heads, tails];
   PrependTo[mat, t /@ tails];
   mat = Prepend[Transpose[mat], Prepend[h /@ heads, ω]];
   MatrixForm[mat]
  ];
R[x_, y_] := B[1, (T_x - 1) t[x] h[y]];
Rinv[x_, y_] := B[1, (1 / T_x - 1) t[x] h[y]];
tm[x_, y_, z_][β_] := β /. {t[x] → t[z], t[y] → t[z], T_x → T_z, T_y → T_z};
hm[x_, y_, z_][B[ω_, μ_]] := Module[
   {γx = D[μ, h[x]], γy = D[μ, h[y]], M = μ /. h[x] | h[y] → 0},
   B[ω, M + h[z] (γx + γy + (γx /. t[i_] :> 1) γy)] // βCollect
  ];
swap[x_, y_][B[ω_, μ_]] := Module[
   {α, β, γ, δ, ϵ},
   α = Coefficient[μ, h[x] t[y]];
   β = D[μ, t[y]] /. h[x] → 0;
   γ = D[μ, h[x]] /. t[y] → 0;
   δ = μ /. h[x] | t[y] → 0;
   ϵ = 1 + α;
   B[ω * ϵ, Plus[
      α (1 + (γ /. t[i_] :> 1) / ϵ) h[x] t[y],
      β (1 + (γ /. t[i_] :> 1) / ϵ) t[y],
      γ / ϵ h[x],
      δ - (1 / ϵ) γ * β
     ]] // βCollect
  ];
gm[x_, y_, z_][β_] := β // swap[y, x] // hm[x, y, z] // tm[x, y, z];
B /: B[ω1_, μ1_] B[ω2_, μ2_] := B[ω1 * ω2, μ1 + μ2];
```

```
βMVA[L_Link, shifts___] := Module[
  {skel, β, s, k, sh, vars},
  skel = Skeleton[L];
  vars = First /@ skel;
  sh = PadRight[{shifts}, Length[skel]];
  skel = MapThread[RotateLeft[#2, #1] &, {sh, skel}];
  β = Times @@ GD[L] /. {Ar[x_, y_, +1] :> R[x, y], Ar[x_, y_, -1] :> Rinv[x, y]};
  Do[
   Do[
    β = β // gm[skel[[s, 1]], skel[[s, k]], skel[[s, 1]]],
    {k, 2, Length[skel[[s]]]}
   ],
   {s, Length[skel]}
  ];
  β /. Flatten[
    MapThread[{T#1 → X#2, t[#1] → t[#2], h[#1] → h[#2]} &, {First /@ skel, vars}]]
 ]
```

```
{L = Link["L8n1"]; βMVA[L], βMVA[L, 0, 1], βMVA[L, 2, 2]} /.
  B[ω_, μ_] :> B[ω, ω * μ] // βForm
```

KnotTheory::loading : Loading precomputed data in PD4Links`.

$$\left\{\begin{pmatrix} -\frac{1-3X_5+X_5^2}{X_5^2} & h[1] & h[5] \\ t[1] & -\frac{(-1+X_1)(-1+X_5)(-1+2X_5)}{X_1X_5^4} & \frac{(-1+X_1)\left(1-2X_5-2X_1X_5^2+X_1X_5^3\right)}{X_1^2X_5^3} \\ t[5] & \frac{(-1+X_5)\left(1-2X_5-2X_1X_5^2+X_1X_5^3\right)}{X_1X_5^4} & \frac{(-1+X_1)(-1+X_5)\left(1-2X_5-X_1X_5+X_5^2\right)}{X_1^2X_5^3} \end{pmatrix},\right.$$

$$\begin{pmatrix} -\frac{1-X_5-2X_1X_5+X_1X_5^2}{X_1X_5} & h[1] & h[5] \\ t[1] & \frac{(-1+X_1)(-2+X_5)(-1+X_5)}{X_1X_5^2} & \frac{(-1+X_1)\left(1-2X_5-2X_1X_5^2+X_1X_5^3\right)}{X_1^2X_5^3} \\ t[5] & \frac{(-1+X_5)\left(1-2X_5-2X_1X_5^2+X_1X_5^3\right)}{X_1X_5^3} & \frac{(-1+X_1)(-1+X_5)\left(X_1-X_1X_5-X_5^2-2X_1X_5^2-2X_1^2X_5^2+X_1X_5^3+X_1^2X_5^3\right)}{X_1^3X_5^3} \end{pmatrix},$$

$$\left.\begin{pmatrix} -\frac{1-3X_5+3X_5^2-2X_1X_5^2-X_5^3+X_1X_5^3}{X_1X_5} & h[1] & h[5] \\ t[1] & \frac{(-1+X_1)(-1+X_5)\left(1-2X_5-2X_1X_5^2-X_1X_5^3+X_1X_5^4\right)}{X_1^2X_5^3} & \frac{(-1+X_1)\left(1-2X_5-2X_1X_5^2+X_1X_5^3\right)}{X_1^2X_5} \\ t[5] & \frac{(-1+X_5)\left(1-2X_5-2X_1X_5^2+X_1X_5^3\right)}{X_1^2X_5^3} & \frac{(-1+X_1)(-1+X_5)\left(-1+2X_5+X_1X_5-X_5^2\right)}{X_1^3X_5} \end{pmatrix}\right\}$$

```
{L = Link["L8a15"]; βMVA[L], βMVA[L, 0, 1, 1], βMVA[L, 2, 3, 1]} /.
   B[ω_, μ_] :→ B[ω, ω * μ] // βForm // ColumnForm
```

$$\begin{pmatrix} -\frac{(-1+X_1+X_5)\,(-1+X_1+X_9)\,(2-2X_5-2X_9+X_5X_9)}{X_1^2X_5^2X_9} & h[1] & \\ t[1] & \frac{(-1+X_1)\,(2-2X_5-2X_9+X_5X_9)\,(1-X_1-X_5-X_9+X_5X_9+X_1X_5X_9)}{X_1^2X_5^3X_9^2} & \\ t[5] & \frac{(-1+X_5)\,\left(-1+2X_1+2X_5-2X_1X_5+2X_9-2X_1X_9-3X_5X_9+X_1X_5X_9-X_9^2+X_5X_9^2\right)}{X_1^2X_5^2X_9} & \frac{(-1+X_5)\,\left(2-4X_1+2\cdots\right.}{} \\ t[9] & \frac{(-1+X_9)\,(-1+2X_1+X_5-2X_1X_5+X_9-2X_1X_9-X_5X_9+X_1X_5X_9)}{X_1^2X_5^2X_9} & \end{pmatrix}$$

$$\begin{pmatrix} -\frac{2-2X_5-2X_9+X_5X_9}{X_5} & h[1] & h[5] & \\ t[1] & 0 & \frac{(-1+X_1)\,(2-2X_5-2X_9+X_5X_9)}{X_1X_5^2X_9} & \\ t[5] & \frac{(-1+X_5)\,(2-2X_5-2X_9+X_5X_9)}{X_5^2} & \frac{(-1+X_5)\,\left(2X_5-2X_9+2X_1X_9-X_5X_9-2X_1X_5X_9+2X_9^2-2X_1X_9^2+X_1X_5X_9^2\right)}{X_1X_5^2X_9} & \\ t[9] & \frac{(-1+X_9)\,(2-2X_5-2X_9+X_5X_9)}{X_5^2X_9} & \frac{(-1+X_9)\,(-2+2X_1+2X_5-2X_1X_5+2X_9-2X_1X_9-2X_5X_9+X_1X_5X_9)}{X_1X_5^2X_9} & \frac{(-1+X_9)\,\left(-2X_5+2\cdots\right.}{} \end{pmatrix}$$

$$\begin{pmatrix} -\frac{(-1+X_1+X_5)\,X_9\,(1-2X_5-X_9+X_5X_9)}{X_1X_5^2} & h[1] & h[5] \\ t[1] & \frac{(-1+X_1)\,(-1+X_5)\,(1-2X_5-X_9+X_5X_9)}{X_1X_5^3} & \frac{(-1+X_1)\,(1-2X_5-X_9+X_5X_9)}{X_1X_5^2} \\ t[5] & \frac{(-1+X_5)\,(1-2X_5-X_9+X_5X_9)}{X_1X_5^2} & \frac{(-1+X_5)\,\left(-1-X_1+2X_1^2+2X_5+X_9+2X_1X_9-3X_1^2X_9-X_5X_9-2X_1X_5X_9\cdots\right.}{X_1^2X_5^2} \\ t[9] & \frac{(-1+X_1+X_5)\,(-1+X_9)\,(1-2X_5-X_9+X_5X_9)}{X_1X_5^2} & -\frac{(-1+X_1+X_5)\,(-1+X_9)}{X_1X_5^2} \end{pmatrix}$$